# -------------------------------------------------
# Project created by QtCreator 2010-05-03T16:13:43
# -------------------------------------------------
QT += network \
	sql \
	xml \
	webkit
greaterThan(QT_MAJOR_VERSION, 4): QT += widgets webkitwidgets

CONFIG += uitools
CONFIG += t_core t_yt_dump_lib
TARGET = yt_new

TEMPLATE = app

# ###CONFIG += link_prl
# static plugin
contains( DEFINES, QT_STATIC_BUILD ) {
    # !contains( DEFINES, QT_DLL ) {
    QTPLUGIN += qgif \
        qico \
        qjpeg \
        qmng
    QTPLUGIN += qcncodecs \
        qjpcodecs \
        qkrcodecs \
        qtwcodecs
    #QTPLUGIN += qsqlodbc \
    #    odbc32
}

# ###    LIBS += -lodbc32 -lodbccp32
SOURCES += main.cpp \
    widget.cpp \
    xmlutil.cpp \
    yt_rec.cpp \
    yt_web.cpp \ # mediadata.cpp \
    webview.cpp \
    mainwindow.cpp \
    env.cpp \
    pages.cpp \
    batchnetworkaccessmanager.cpp
#    t_box_api_backend.cpp
HEADERS += widget.h \
    xmlutil.h \
    yt_rec.h \
    yt_web.h \
    webview.h \
    mainwindow.h \
    env.h \
    pages.h \
    batchnetworkaccessmanager.h
#    t_box_api_backend.h
FORMS += widget.ui \
    mainwindow.ui
RESOURCES += yt_dump.qrc
OTHER_FILES += yt_list.js \
    page/page1.html \
    page/page2.html \
    translations/qt_ja_JP.ts \
    page/page2.list1.updated.js \
    page/page2_list1_updated.js \
    yt_list.css \
    yt_new.rc \
    jslib/jquery.timers-1.2.js \
    jslib/jquery-1.8.2.min.js \
    jslib/dateformat.js \
    jslib/json2.js

RC_FILE = yt_new.rc

#[export]
build_pass:CONFIG(debug, debug|release) {
    TARGET = $$join(TARGET,,,_debug)
}
CONFIG(release, debug|release) {
    target.path = C:/usr/local/release
    INSTALLS += target
} else {
    target.path = C:/usr/local/debug
    INSTALLS += target
}

# http://www.qtcentre.org/wiki/index.php?title=Undocumented_qmake
# [lupdate]
# LUPDATE_FILES += yt_dump.pro
win32:QMAKE_LUPDATE = $$[QT_INSTALL_BINS]\\lupdate.exe
else:QMAKE_LUPDATE = $$[QT_INSTALL_BINS]/lupdate
LUQM.depends = $$SOURCES \
    $$HEADERS \
    $$FORMS
LUQM.input = LUPDATE_FILES
LUQM.output = translations/yt_dump_ja_JP.ts
LUQM.commands = $$QMAKE_LUPDATE \
    yt_dump.pro \
    -ts \
    translations/yt_dump_ja_JP.ts
LUQM.CONFIG = no_link
QMAKE_EXTRA_COMPILERS += LUQM
#PRE_TARGETDEPS += compiler_LUQM_make_all

# [lrelease]
LRELEASE_FILES += translations/qt_ja_JP.ts \
    translations/yt_dump_ja_JP.ts
win32:QMAKE_LRELEASE = $$[QT_INSTALL_BINS]\\lrelease.exe
else:QMAKE_LRELEASE = $$[QT_INSTALL_BINS]/lrelease
RLQM.name = lrelease \
    ${QMAKE_FILE_IN}
RLQM.input = LRELEASE_FILES
RLQM.output = translations/${QMAKE_FILE_BASE}.qm
RLQM.commands = $$QMAKE_LRELEASE \
    ${QMAKE_FILE_IN}
RLQM.CONFIG = no_link
QMAKE_EXTRA_COMPILERS += RLQM
#PRE_TARGETDEPS += compiler_RLQM_make_all
